

# Multi-Year Production and Facility Optimization Report

SCH-MGMT 752 – Business Process Optimization

Yash Shah & Peiyao Zhao

November 2025

# Contents

<b>1 Executive Summary</b>	<b>3</b>
<b>2 Context and Scope of Work</b>	<b>3</b>
<b>3 Methodology – Modeling and Optimization</b>	<b>4</b>
3.1 Optimal Model Development . . . . .	4
3.2 Scenarios Analysis . . . . .	5
<b>4 Findings and Analysis</b>	<b>6</b>
4.1 Optimized Shipments Across Plants and Markets . . . . .	6
4.1.1 Capacity Utilization Under the Optimized Scenario . . . . .	6
4.1.2 Summary of Optimized Solutions . . . . .	9
4.2 Profit Comparison Across Scenarios . . . . .	9
<b>5 Recommendations</b>	<b>10</b>
<b>A Code Sources</b>	<b>11</b>

# 1 Executive Summary

This report presents a mixed-integer optimization-based consulting analysis for a manufacturing company producing four products — Alpha, Beta, Gamma, and Delta — over a three-year planning horizon. There are two plants under consideration. The first is the Chicago plant, which is already in operation and can be expanded if needed. The second is a proposed new plant in Sunnyvale, California, which can be opened in any year and be expanded depending on strategic decisions. The objective is to maximize total profit by deciding:

- When (if ever) to open the California plant.
- Whether and when to expand Chicago.
- Whether to invest in Delta development in year 1.
- How to allocate production and shipments across plants, markets, and years.

The model accounts for detailed test-line and assembly-hour capacities, year-specific demand ceilings, shipping costs, and one-time fixed decisions (open, expand, develop new products Delta).

## Key results:

- The mixed-integer model identifies the optimal sequence of capacity investments and annual production plans that increase total 3-year profit by approximately 35% compared to the baseline.
- Chicago's expansion relieves the C-line bottleneck; the California plant opens in Year 1 to exploit lower shipping costs to western markets.
- Delta development is rejected because its marginal contribution fails to offset its Year-1 fixed cost.

# 2 Context and Scope of Work

## Business Context

The company must plan production and market distribution for four products sold in three major regions (Chicago, California, Seattle). Each product requires:

- One type of testing (A or C),
- Assembly labor, and
- Per-unit manufacturing and shipping costs.

The two plants differ in capacity, cost, and location.

## Strategic Decisions

Decision	Options	Timing
California Plant Opening	Year 1–3 (or never)	Once opened, stays open
Chicago Expansion	Year 1–3 (or never)	Adds test + assembly capacity
Delta Development	Optional (Year 1 only)	Enables Delta production thereafter

## Scope

- **In-scope:** production planning, plant timing, shipping allocation, and profit optimization.
- **Out-of-scope:** pricing strategy, labor hiring, inventory carry-over, or new product R&D after Delta.

### 3 Methodology – Modeling and Optimization

#### 3.1 Optimal Model Development

##### Model Parameters

Symbol	Description
$y$	Planning horizon (years): years 1–3
$k$	Set of potential plants: Chicago, Sunnyvale
$p$	Set of products: Alpha (A), Beta (B), Gamma (G), Delta (D)
$R_{p,m,y}^k$	Revenue per unit of product $p$ made at plant $k$ and marketed in region $m$ in year $y$ (\$)
$UC_p^k$	Unit production cost for product $p$ at plant $k$ (\$) per unit
$SC_{p,m}^k$	Unit shipping cost of product $p$ to market from plant $k$ $m$ (\$) per unit
$FC_k$	Fixed cost of developing product Delta at plant $k$ (\$ million/year)
$OC_s$	One-time opening (construction) cost of Sunnyvale plant (\$ million)
$EC_c$	One-time expansion cost of Chicago plant (\$ million)
$\alpha_p^k$	Annual capacity of plant $k$ (in units)
$D_{p,m,y}$	Demand for product $p$ in market $m$ in year $y$ (units)
$P_\ell$	Set of products processed on testing line $\ell$ . $P_1 = \{A, B\}$ for the A-line (Alpha, Beta), $P_2 = \{G, D\}$ for the C-line (Gamma, Delta).
$Base_\ell^k$	Base testing capacity of line $\ell$ at plant $k$ (e.g., 6000 for Chicago A-line).
$\gamma_\ell^k$	Per-unit expansion multiplier for line $\ell$ at plant $k$ (e.g., 2000 for Chicago A-line).
$AssyBase^k$	Base assembly capacity at plant $k$ (e.g., 100,000 for Chicago).
$AssyGrow^k$	Assembly capacity added per open/expansion decision at plant $k$ (e.g., 33,000 for Chicago).

##### Decision Variables

$$\begin{aligned}
x_{p,m,y}^k &\geq 0 && \text{Units of product } p \text{ shipped from plant } k \in \{C, S\} \text{ to market } m \text{ in year } y \\
E_y^{C,exp} &\in \{0, 1\} && 1 \text{ if Chicago expansion occurs in year } y \\
O_y^{S,open} &\in \{0, 1\} && 1 \text{ if Sunnyvale is opened in year } y \\
z_k^\Delta &\in \{0, 1\} && 1 \text{ if Delta is developed at plant } k \text{ in Year 1, } k \in \{\text{Chicago, Sunnyvale}\} \quad (1)
\end{aligned}$$

##### Objective Function

$$\begin{aligned}
\max_{x, y, z} \Pi = & \sum_{y=1}^3 \sum_p \sum_m \sum_{k \in \{C, S\}} \left[ (R_{p,m,y}^k - UC_p^k - SC_{p,m}^k) x_{p,m,y}^k \right] \\
& - \sum_{k \in \{C, S\}} z_k^\Delta FC_k - \sum_{y=1}^3 E_y^{C,exp} EC_c - \sum_{y=1}^3 O_y^{S,open} OC_s \quad (2)
\end{aligned}$$

##### Constraints

###### (1) Demand Constraints:

$$\sum_{k \in \{C, S\}} x_{p,m,y}^k \leq D_{p,m,y}, \quad \forall p, m, y \quad (3)$$

(2) Testing Capacity (A-line for Alpha, Beta; C-line for Gamma, Delta):

$$\sum_m \sum_{p \in P^\ell} x_{p,m,y}^k \leq \text{Base}_\ell^k + \gamma_\ell^k \sum_{\tau=1}^y E_\tau^{k,\exp} \quad \forall k, \forall \ell, \forall y \quad (4)$$

(3) Assembly Capacity:

$$\sum_p \sum_m \alpha_p^k x_{p,m,y}^k \leq \text{AssyBase}^k + \text{AssyGrow}^k \sum_{\tau=1}^y O_y^{S,\text{open}} \quad \forall k, \forall y \quad (5)$$

(4) Delta Production:

$$x_{D,m,y}^k \leq z_k^\Delta D_{D,m,y}, \quad \forall m, y \quad (6)$$

(7)

(5) Plant Opening Logic:

$$x_{p,m,y}^S \leq D_{p,m,y} \sum_{\tau=1}^y O_y^{S,\text{open}}, \quad \forall p, m, y \quad (8)$$

(6) Integrality and Non-negativity:

$$x_{p,m,y}^k \geq 0; \quad E_y^{C,\exp}, O_y^{S,\text{open}}, z_k^\Delta \in \{0, 1\} \quad (9)$$

### 3.2 Scenarios Analysis

The model is solved under multiple scenarios to evaluate the impact of strategic decisions. We consider five operational scenarios to evaluate production, shipment, and capacity utilization:

1. **Baseline (Chicago Only):** Only the Chicago plant operates and no expansion in Chicago plan. All shipments originate from Chicago, increasing transportation costs to distant markets and limiting the ability to meet demand, particularly for higher-margin products.
2. **Optimal Model:** Chicago is expanded and a new plant is established, with optional Delta product development. This scenario might allow for better market coverage and efficient capacity utilization.
3. **No Expansion:** A new plant is opened, but Chicago is not expanded. This isolates the effect of adding a new facility without modifying existing capacity.
4. **No New Plant:** Chicago expands, but no additional plant is added, relieving local bottlenecks but leaving distant markets constrained.
5. **Force Delta Development:** Combines plant expansion, a new plant, and mandatory Delta product development to examine the maximum achievable production and sales.

More details about each scenario are shown in Table 2:

Table 2: Scenario Configuration Summary

Scenario	Plant Expansion	New Plant	Delta Development
Baseline	-	-	-
Optimal Model	✓	✓	optional
No Expansion	-	✓	-
No New Plant	✓	-	-
Force Delta Development	✓	✓	✓

## 4 Findings and Analysis

### 4.1 Optimized Shipments Across Plants and Markets

In the optimized scenario, production is allocated to minimize total costs while meeting market demand. Chicago primarily serves the Chicago market, while Sunnyvale plant serves California and Seattle markets, reducing shipping distances and costs. Delta shipments remain zero because developing Delta is not profitable. The number of the products shipped from different plants to various markets is shown in Table 3. The actual product sales (shipped) to market of each product versus the forecast maximum sales is shown in ???. The shipment plan demonstrates several key points:

- Sunnyvale serves most of the CA and SEA demand, significantly reducing shipping cost.
- Chicago focuses primarily on Chicago market, plus some SEA flows where still profitable.
- The split between Chicago and Sunnyvale allows the firm to sell more total volume in profitable segments while staying within assembly and testing capacities.
- Delta remains at 0 shipments because its margin minus fixed development cost is not attractive compared to focusing capacity on Gamma and the other products.




Figure 1: The actual sales of each product versus the forecast maximum sales.

#### 4.1.1 Capacity Utilization Under the Optimized Scenario

Capacity utilization indicates where production limits are binding and highlights bottlenecks. This section shows how much of each key capacity is used under the optimized solution. The numbers are aggregate across all products and markets for that resource.

Table 3: Optimized Shipments by Year, Plant, Product, and Market in the optimal model

Year	Plant	Product	Market	Quantity
1	Chicago	Alpha	Chicago	3000
	Chicago	Beta	Chicago	2000
	Chicago	Gamma	Chicago	1711
	Sunnyvale	Alpha	CA	1500
	Sunnyvale	Alpha	SEA	1200
	Sunnyvale	Beta	CA	1000
	Sunnyvale	Beta	SEA	1000
	Sunnyvale	Gamma	Chicago	289
	Sunnyvale	Gamma	CA	500
	Sunnyvale	Gamma	SEA	750
2	Chicago	Alpha	Chicago	6000
	Chicago	Beta	Chicago	1000
	Chicago	Beta	SEA	300
	Chicago	Gamma	Chicago	2000
	Chicago	Gamma	SEA	500
	Sunnyvale	Alpha	CA	2000
	Sunnyvale	Alpha	SEA	1800
	Sunnyvale	Beta	CA	500
	Sunnyvale	Beta	SEA	200
	Sunnyvale	Gamma	CA	1000
3	Sunnyvale	Gamma	SEA	1000
	Chicago	Alpha	Chicago	3000
	Chicago	Beta	Chicago	2500
	Chicago	Beta	SEA	843
	Chicago	Gamma	Chicago	2000
	Chicago	Gamma	SEA	500
	Sunnyvale	Alpha	CA	1000
	Sunnyvale	Alpha	SEA	1000
	Sunnyvale	Beta	CA	1500
	Sunnyvale	Beta	SEA	357
	Sunnyvale	Gamma	CA	1000
	Sunnyvale	Gamma	SEA	1000

As shown in Table 4 and Figure 2, the C-line is not fully saturated and there is slack in every year, indicating that testing Gamma/Delta is not the critical bottleneck in Chicago. As for A-Line, as shown in Table 5 and Figure 3, the A-line is heavily used in Year 2, but some slack remains. Testing is not fully constraining production in any year.

Table 4: Chicago C-Line Capacity Utilization (Gamma/Delta Testing)

Year	Used Hours	Capacity Hours
1	1711	3200
2	2500	3200
3	2500	3200

Table 5: Chicago A-Line Capacity Utilization (Alpha/Beta Testing)

Year	Used Hours	Capacity Hours
1	5000	8000
2	7300	8000
3	6342.9	8000

Further, on the assembly line, assembly at Chicago becomes very tight in Years 2 and 3, with only a few thousand hours of slack (shown in Table 6). This indicates assembly, rather than testing, is the main

capacity pressure at Chicago. On the other hand, Sunnyvale plant assembly is fully utilized in every year, as illustrated in Table 7. This is a true bottleneck, which indicates that once Sunnyvale is full, any additional West Coast demand has to be served from Chicago at higher cost (and with limited Chicago assembly headroom).




Figure 2: Testing Line Capacity Utilization for Chicago A and C lines.

Table 6: Chicago Assembly Capacity Utilization

Year	Used Hours	Capacity Hours
1	94222.2	133000
2	129500	133000
3	130142.9	133000

Table 7: Sunnyvale Assembly Capacity Utilization

Year	Used Hours	Capacity Hours
1	80000	80000
2	80000	80000
3	80000	80000




Figure 3: Assembly Line Capacity Utilization for Chicago and Sunnyvale lines.

#### 4.1.2 Summary of Optimized Solutions

Overall, the optimized solution strategically balances production between Chicago and Sunnyvale, fully utilizes available capacities, minimizes shipping costs, and maximizes total profit while avoiding unprofitable Delta development. Insights from this optimal model results are summarized in Table 8.

Table 8: Summary of Key Insights from the Optimized Solution

Category	Key Insights
Profitability	Baseline profit: \$10.182M. Optimized profit: \$19.187M. Profit increases by approximately \$9.0M (~ 88%) due to opening Sunnyvale in Year 1 and expanding Chicago in Year 1.
Role of Sunnyvale Plant	Sunnyvale serves CA and SEA markets with lower shipping cost. Sunnyvale assembly is fully saturated in all three years, indicating it is a binding capacity constraint.
Chicago Expansion	Chicago expansion increases testing and assembly capacity. Chicago assembly becomes tight in Years 2 and 3 but does not fully saturate. Chicago testing lines (A-line and C-line) remain underutilized, indicating they are not bottlenecks.
Delta Product	Delta is never produced in the optimal solution. Its margin minus development cost is inferior to producing Gamma and other products. Delta development is not economically justified under current parameters.
Demand	Demand is not fully met due to limited assembly capacity, especially at Sunnyvale. The model allocates capacity to the most profitable product–market combinations.

## 4.2 Profit Comparison Across Scenarios

We first compare total 3-year profit under different strategic scenarios. This allows us to quantify the impact of opening a new plant, expanding existing capacity, and developing the Delta product. The comparison of the profit for each scenario is shown in Table 9.

The optimized solution almost doubles total profit ( $\approx +88\%$ ) relative to the baseline. The main drivers of increased profit are:

- Opening the Sunnyvale plant in California in Year 1, which reduces shipping costs and better serves West Coast markets.

Table 9: Profit Comparison by Scenario

Scenario	Profit (USD)
Baseline	10,182,000
Optimal Model	19,186,900
No Expansion	18,212,300
No New Plant	12,578,100
Force Delta Development	18,994,600

- Expanding Chicago (Chicago) in Year 1, which alleviates assembly bottlenecks and allows higher production of high-margin products.
- Strategic decision to not develop Delta, as its expected contribution does not cover the fixed development cost.

## 5 Recommendations

Based on the optimized solution and analysis, we provide the following recommendations:

- **Open Sunnyvale in Year 1 and maintain it throughout the planning horizon.**
  - Sunnyvale is essential: its assembly capacity is fully utilized every year and it enables profitable coverage of the California and Seattle markets.
- **Expand Chicago in Year 1 rather than delaying expansion.**
  - Early expansion unlocks extra capacity across all three years, allowing more profitable production in Years 1–3 instead of only later years.
- **Treat Sunnyvale as the primary West Coast hub.**
  - Use Sunnyvale to serve CA and SEA whenever capacity allows. Chicago should primarily serve Chicago and selective SEA flows as backup.
- **Do not develop Delta under current conditions.**
  - Delta’s economics do not justify the fixed development cost plus usage of scarce capacity.
  - Only reconsider Delta development if:
    - \* Delta’s price increases,
    - \* Delta’s manufacturing or shipping cost decreases, or
    - \* The fixed development cost drops substantially.
- **Future capacity planning should prioritize assembly over testing.**
  - Both Chicago and Sunnyvale assembly are or become tight, while Chicago test lines have spare capacity.
  - Any future investment should focus on adding assembly capacity, particularly at Sunnyvale.

## A Code Sources

### Model

```
# =====
# SCH-MGMT 752 - Team Challenge
# Multi-year, multi-plant, multi-product production model
# - Existing plant: Chicago (CHI), always open
# - Optional new plant: Sunnyvale, CA (NEW), open in at most one year
# - Optional expansion of CHI, at most once (any year)
# - Optional Delta development at at most one plant (CHI or NEW)
# - 3 years, 3 markets, 4 products
# Objective: maximize total 3-year profit
# =====

#####
# 1. SETS
#####

set P;                      # Products: Alpha, Beta, Gamma, Delta
set M;                      # Markets: CHI, CA, SEA
set K;                      # Plants: CHI (existing), NEW (Sunnyvale)
set Y;                      # Years: 1,2,3
set L;                      # Test lines: A-line, C-line

#####
# 2. PARAMETERS
#####

# Demand upper bounds: max sales per product, market, year
param D{P,M,Y} >= 0;

# Revenue per unit for each product (same across markets/years in the PDF)
param R{P} >= 0;

# Manufacturing cost per unit at plant k
param C_mfg{P,K} >= 0;

# Shipping cost per unit from plant k to market m
param C_ship{P,M,K} >= 0;

# --- Testing time per unit on each line ---
# A-line: Alpha & Beta need 1 hour, others 0
# C-line: Gamma & Delta need 1 hour, others 0
param a_test{P,L} >= 0;

# Assembly hours per unit of each product
param a_assy{P,K} >= 0;

# --- Chicago capacities (base + extra if expanded) ---
# Base test capacity per line at CHI (same each year)
param Tbase_CHI{L} >= 0;          # e.g., A:6000, C:2400
param Texp_CHI{L}  >= 0;          # extra after expansion (e.g., A:+2000, C:+800)
```

```

# Base assembly capacity at CHI and extra after expansion
param Sbase_CHI >= 0;                      # e.g., 100000
param Sexp_CHI  >= 0;                      # e.g., 33000

# --- New plant capacities (only if open) ---
# These are per year when the NEW plant is open
param T_NEW{L} >= 0;                      # e.g., A:5000, C:2000
param S_NEW    >= 0;                      # e.g., 80000

# --- Fixed costs ---
param F_open_NEW  >= 0;                  # fixed cost to open NEW plant (once)
param F_exp_CHI   >= 0;                  # fixed cost for CHI expansion (once)
param F_dev_Delta{K} >= 0;                # fixed cost to develop Delta at plant k

#####
# 3. DECISION VARIABLES
#####

# Shipments: units of product p to market m in year y from plant k
var x{P,M,Y,K} >= 0;

# Plant open status: 1 if plant k is open in year y
var o{K,Y} binary;

# New-plant opening event: 1 if NEW is opened in year y (at most one y)
var u_open_NEW{Y} binary;

# Chicago expansion event: 1 if CHI is expanded in year y (at most one y)
var e_CHI{Y} binary;

# Cumulative expansion indicator: 1 if CHI is expanded in or before year y
var E_CHI{Y} binary;

# Delta development decision: 1 if Delta is developed at plant k
var dDelta{K} binary;

#####
# 4. DERIVED PARAMETERS
#####

# Per-unit net margin (revenue - mfg - shipping) at market m from plant k
param margin{p in P, m in M, k in K} :=
  R[p] - C_mfg[p,k] - C_ship[p,m,k];

#####
# 5. OBJECTIVE: MAX PROFIT
#####

maximize Profit:
  # Variable margin from all shipments
  sum{p in P, m in M, y in Y, k in K} margin[p,m,k] * x[p,m,y,k]
  # Fixed costs: open NEW, expand CHI, develop Delta
  - sum{y in Y} F_open_NEW      * u_open_NEW[y]
  - sum{y in Y} F_exp_CHI     * e_CHI[y]

```

```

- sum{k in K} F_dev_Delta[k]* dDelta[k] ;

#####
# 6. CONSTRAINTS
#####

# 6.1 Chicago is always open
s.t. CHIAwaysOpen{y in Y}:
    o["CHI",y] = 1;

# 6.2 New plant open status propagation (open at most once)
s.t. NEW_OpenOnce:
    sum{y in Y} u_open_NEW[y] <= 1;

s.t. NEW_OpenStatus{y in Y}:
    o["NEW",y] = sum{t in Y: t <= y} u_open_NEW[t];

# 6.3 Chicago expansion cumulative (expand at most once)
s.t. CHI_AtMostOneExp:
    sum{y in Y} e_CHI[y] <= 1;

s.t. CHI_CumExp{y in Y}:
    E_CHI[y] = sum{t in Y: t <= y} e_CHI[t];

# 6.4 At most one plant can develop Delta
s.t. Delta_OnePlant:
    sum{k in K} dDelta[k] <= 1;

# 6.5 Testing capacity constraints

# A-line at CHI and NEW:
s.t. TestCap_A{y in Y, k in K}:
    sum{p in P, m in M} a_test[p,"A"] * x[p,m,y,k]
    <= (
        if k = "CHI" then
            (Tbase_CHI["A"] + Texp_CHI["A"] * E_CHI[y])
        else
            T_NEW["A"]
    ) * o[k,y];

# C-line at CHI and NEW:
s.t. TestCap_C{y in Y, k in K}:
    sum{p in P, m in M} a_test[p,"C"] * x[p,m,y,k]
    <= (
        if k = "CHI" then
            (Tbase_CHI["C"] + Texp_CHI["C"] * E_CHI[y])
        else
            T_NEW["C"]
    ) * o[k,y];

# 6.6 Assembly capacity constraints

s.t. AssyCap{y in Y, k in K}:
    sum{p in P, m in M} a_assy[p,k] * x[p,m,y,k]

```

```

<= (
    if k = "CHI" then
        (Sbase_CHI + Sexp_CHI * E_CHI[y])
    else
        S_NEW
    ) * o[k,y];

# 6.7 Demand / sales upper bounds

s.t. SalesCap{p in P, m in M, y in Y}:
    sum{k in K} x[p,m,y,k] <= D[p,m,y];

# 6.8 Ship only if plant is open (big-M using D as upper bound)

s.t. ShipIfOpen{p in P, m in M, y in Y, k in K}:
    x[p,m,y,k] <= D[p,m,y] * o[k,y];

# 6.9 Delta gating: only if developed at that plant, and plant is open

s.t. DeltaGate{m in M, y in Y, k in K}:
    x["Delta",m,y,k] <= D["Delta",m,y] * dDelta[k] * o[k,y];

# (Optional: if you want to explicitly enforce "decision in year 1" for Delta dev,
# you could add a comment or constraint linking dDelta to year 1 status, but
# economically it's not necessary since dDelta has no year index.)

```

## Data

```

# =====
# Data for Team Challenge 1
# =====

set P := Alpha Beta Gamma Delta;
set M := CHI CA SEA;
set K := CHI NEW;
set Y := 1 2 3;
set L := A C;

# ----- Revenues (from case text) -----
param R :=
    Alpha 1350
    Beta 1650
    Gamma 3000
    Delta 2500
;

# ----- Demand D[p,m,y] (Table 1 in the PDF) -----
# PLEASE fill these values exactly from the PDF.
# Example layout (numbers below are placeholders-replace with real ones):

param D :=
# Year 1
[Alpha, CHI, 1] 3000 [Alpha, CA, 1] 1500 [Alpha, SEA, 1] 1200
[Beta, CHI, 1] 2000 [Beta, CA, 1] 1000 [Beta, SEA, 1] 1000

```

```

[Gamma, CHI, 1] 2000 [Gamma, CA, 1] 500 [Gamma, SEA, 1] 750
[Delta, CHI, 1] 500 [Delta, CA, 1] 300 [Delta, SEA, 1] 300

# Year 2
[Alpha, CHI, 2] 6000
[Alpha, CA, 2] 2000
[Alpha, SEA, 2] 1800
[Beta, CHI, 2] 1000
[Beta, CA, 2] 500
[Beta, SEA, 2] 500
[Gamma, CHI, 2] 2000
[Gamma, CA, 2] 1000
[Gamma, SEA, 2] 1500
[Delta, CHI, 2] 1000
[Delta, CA, 2] 600
[Delta, SEA, 2] 600

# Year 3
[Alpha, CHI, 3] 3000
[Alpha, CA, 3] 1000
[Alpha, SEA, 3] 1000
[Beta, CHI, 3] 2500
[Beta, CA, 3] 1500
[Beta, SEA, 3] 1200
[Gamma, CHI, 3] 2000
[Gamma, CA, 3] 1000
[Gamma, SEA, 3] 1500
[Delta, CHI, 3] 2000
[Delta, CA, 3] 1500
[Delta, SEA, 3] 1200
;

# ----- Manufacturing costs C_mfg[p,k] -----
# From the case text where it lists cost at CHI vs NEW.
param C_mfg :=
    [Alpha, CHI] 1000 [Alpha, NEW] 925
    [Beta, CHI] 1175 [Beta, NEW] 1100
    [Gamma, CHI] 2250 [Gamma, NEW] 2125
    [Delta, CHI] 2100 [Delta, NEW] 1900
;

# ----- Shipping costs C_ship[p,m,k] -----
# From the 2 shipping tables in the case (check carefully).

param C_ship :=
# From CHI plant
[Alpha, CHI, CHI] 22 [Alpha, CA, CHI] 52 [Alpha, SEA, CHI] 50
[Beta, CHI, CHI] 19 [Beta, CA, CHI] 48 [Beta, SEA, CHI] 46
[Gamma, CHI, CHI] 27 [Gamma, CA, CHI] 58 [Gamma, SEA, CHI] 56
[Delta, CHI, CHI] 27 [Delta, CA, CHI] 58 [Delta, SEA, CHI] 56

# From NEW plant
[Alpha, CHI, NEW] 72 [Alpha, CA, NEW] 20 [Alpha, SEA, NEW] 30
[Beta, CHI, NEW] 48 [Beta, CA, NEW] 17 [Beta, SEA, NEW] 26

```

```

[Gamma, CHI, NEW] 58    [Gamma, CA, NEW] 25    [Gamma, SEA, NEW] 35
[Delta, CHI, NEW] 58    [Delta, CA, NEW] 25    [Delta, SEA, NEW] 35
;

# ----- Testing time per unit -----
# A-line: Alpha & Beta need 1 hr, others 0
# C-line: Gamma & Delta need 1 hr, others 0

param a_test :=
  [Alpha, A] 1    [Alpha, C] 0
  [Beta, A] 1    [Beta, C] 0
  [Gamma, A] 0    [Gamma, C] 1
  [Delta, A] 0    [Delta, C] 1
;

# ----- Assembly hours per unit at each plant -----

param a_assy:
  CHI      NEW :=
  Alpha    10     9
  Beta     15     14
  Gamma   20     18
  Delta   22     20
;
;

# ----- Chicago capacities (base + expansion) -----
# Check these against the PDF:
# base A=6000, C=2400; expanded A=8000, C=3200 => extra A=2000, C=800
param Tbase_CHI :=
  A 6000
  C 2400
;
param Texp_CHI :=
  A 2000
  C 800
;
;

# Assembly: base=100000, expanded=133000 => extra=33000
param Sbase_CHI := 100000;
param Sexp_CHI  := 33000;

# ----- New plant capacities (per year when open) -----
# As given in the case (verify):
param T_NEW :=
  A 5000
  C 2000
;
;

param S_NEW := 80000;

# ----- Fixed costs -----
# Opening NEW plant, CHI expansion, Delta development

```

```
param F_open_NEW    := 2225000;
param F_exp_CHI     := 834000;
param F_dev_Delta   :=
    [CHI] 775000
    [NEW] 775000
;
```